# $Id: PKGBUILD 88587 2013-04-20 15:38:12Z jlichtblau $
# Maintainer: Jaroslav Lichtblau <dragonlord@aur.archlinux.org>
# Contributor: Alexander Rødseth <rodseth@gmail.com>

pkgname=nzbget
pkgver=10.2
pkgrel=1
pkgdesc="Download from Usenet using .nzb files"
arch=('x86_64' 'i686')
url="http://nzbget.sourceforge.net/"
license=('GPL')
depends=('gnutls' 'libpar2' 'libxml2' 'libsigc++' 'libgcrypt')
makedepends=('autoconf' 'automake')
changelog=$pkgname.changelog
source=(http://sourceforge.net/projects/$pkgname/files/$pkgname-$pkgver.tar.gz)
sha256sums=('c9b878744c162e7721ffc8b048a2e4550a8ed8da6e706ee4cbd026ed2d612ec2')

build() {
  cd ${srcdir}/$pkgname-$pkgver

  ./configure --prefix=/usr --enable-parcheck
  make CXXFLAGS="-lsigc-2.0"
}

package() {
  cd ${srcdir}/$pkgname-$pkgver

  make DESTDIR=${pkgdir} install

  install -d ${pkgdir}/usr/share/$pkgname
  install -m644 -t ${pkgdir}/usr/share/$pkgname AUTHORS ChangeLog README
}
